科学・技術の世界

数式処理システムによる新時代の数学

算術幾何平均に関する最近の研究

行木 孝夫、松本 圭司 (北大 理 数学)

第１４回, 2018 January 22

算術幾何平均とは

$m_1(x,y)$ を x,y の算術平均(相加平均, the arithmetic mean of x,y) とする．つまり

$$m_1(x,y) = \frac{x+y}{2}.$$

$m_2(x,y)$ を x,y の幾何平均(相乗平均, the geometric mean of x,y) とする．つまり

$$m_2(x,y) = \sqrt{xy}.$$

２つの正数 x,y $(x \geq y)$ に対して，数列 $\{x_n\}$，$\{y_n\}$ を $x_0 = x$，$y_0 = y$ および連立漸化式

$$x_{n+1} = m_1(x_n, y_n) = \frac{x_n + y_n}{2}, \quad y_{n+1} = m_2(x_n, y_n) = \sqrt{x_n y_n}$$

で定義する．

Proposition 1 数列 $\{x_n\}$, $\{y_n\}$ は収束し, $\lim_{n\to\infty} x_n = \lim_{n\to\infty} y_n$.

Proof. 平均の性質および (相加平均) $\geq$ (相乗平均) より

$$y = y_0 \leq y_1 = \sqrt{x_0 y_0} \leq \frac{x_0 + y_0}{2} = x_1 \leq x_0 = x.$$

一般に $y_n \leq x_n$ ならば, $y_n \leq y_{n+1} \leq x_{n+1} \leq x_n$ が成立. 従って,

$$y_0 \leq y_1 \leq \cdots \leq y_n \leq x_n \leq \cdots \leq x_1 \leq x_0$$

をみたし, $\{y_n\}$ は上に有界な単調増加列, $\{x_n\}$ は下に有界な単調減少列なので, これらの数列は収束する.

$\lim_{n\to\infty} x_n = \alpha$, $\lim_{n\to\infty} y_n = \beta$ とおき, 等式 $x_{n+1} = \dfrac{x_n + y_n}{2}$ に対して $n \to \infty$ とすると,

$$\alpha = \lim_{n\to\infty} x_{n+1} = \lim_{n\to\infty} \frac{x_n + y_n}{2} = \frac{\alpha + \beta}{2}$$

となるので, $\alpha = \beta$ となる. $\square$

Definition 1 (参考文献 [1], p.9,10 例３) 正数 x, y $(x \geq y)$ に対して，数列 $\{x_n\}$, $\{y_n\}$ の共通極限

$$\alpha = \lim_{n\to\infty} x_n = \lim_{n\to\infty} y_n$$

を x, y の算術幾何平均 *(the arithmetic-geometric mean of x, y)* といい，$\mathrm{agm}(x, y)$ で表す.

算術幾何平均に関する研究史その１ (高木貞治著 近世数学史談より抜粋)

ガウス(Gauss C.F.) は 1799年(「日記」５月３０日) に 1 と $\sqrt{2}$ との算術幾何平均 $\mathrm{agm}(\sqrt{2}, 1)$ が $\dfrac{\pi}{2\omega}$ と小数第１１位まで一致することを発見した．ここで ω は

$$\omega = \int_0^1 \frac{dx}{\sqrt{1-x^4}}.$$

もしも実際 $\mathrm{agm}(\sqrt{2}, 1) = \dfrac{\pi}{2\omega}$ なることが証明されるならば，「確かに解析の新分野がひらかれるであろう」とガウスは考えている.

このような事実を発見できるのは，天才にしかできないこと．その天才をもってしても，膨大な計算を行い多くの時間を費やしているはず．しかし我々は Maxima を使うと数分でこの事実を確かめることができる．

```
m1(x,y):=(x+y)/2; m2(x,y):=sqrt(x*y);
agm(x,y):=block([i,xx,yy],array(xx,10),array(yy,10),
xx[0]:x,yy[0]:y,
for i from 0 thru 9 do
(xx[i+1]:bfloat(m1(xx[i],yy[i])),
 yy[i+1]:bfloat(m2(xx[i],yy[i])),
 print(i,xx[i],yy[i],xx[i]-yy[i])),
return(xx[10]));

fpprec: 30;
a0:sqrt(2); b0:1;
Gagm:agm(a0,b0);
omega:integrate(1/sqrt(1-x^4),x,0,1);
bfloat(%pi/(2*omega));
bfloat(Gagm-%pi/(2*omega));
```

近世数学史談には

$$\int_0^1 \frac{dx}{\sqrt{1-x^4}} = \int_0^{\pi} \frac{d\varphi}{\sqrt{1-\frac{1}{2}\sin^2\varphi}}$$

とあるが，これは間違い．右辺の積分に $1/(2\sqrt{2})$ をかけないといけない．
正しくは

$$\int_0^1 \frac{dx}{\sqrt{1-x^4}} = \int_0^{\pi} \frac{d\varphi}{2\sqrt{2-\sin^2\varphi}}.$$

```
float(integrate(1/sqrt(1-x^4),x,0,1));
integrate(1/sqrt(1-sin(y)^2/2),y,0,%pi);
romberg(1/sqrt(1-sin(y)^2/2),y,0,%pi);
romberg(1/(2*sqrt(2-sin(y)^2)),y,0,%pi);
```

Maxima において romberg は Riemann 和の近似計算を実行する．

算術幾何平均の計算では，多くの項の計算は必要なくて，x_3，y_3 まで求めればよい．平方根の計算が大変そうであるが，後で紹介する算術調和平均を用いると比較的容易に計算できる．

問題は広義積分の計算にある．この積分は楕円積分とよばれていて，当時は原始関数を見つけて値を求める手段はなかった．オイラー (L. Euler) の先行研究を勉強していたようだ．ベータ関数をガンマ関数で表示して，ガンマ関数の公式を利用したものと思われる．

ガウスは「解析の新分野」と思える新しい理論として楕円関数論、保形関数論を構築している．

算術幾何平均に関する研究史その２

一般の正数 a, b $(a > b)$ に対する $\mathrm{agm}(a, b)$ に対する公式をガウスは構成している．ガウスの超幾何級数 (the hypergeometric series)

$$F(\alpha, \beta, \gamma; z) = \sum_{n=0}^{\infty} \frac{(\alpha, n)(\beta, n)}{(\gamma, n) n!} z^n$$

を用いて表示できる．ここで z が主変数で，α, β, γ はパラメーターで $\gamma \neq 0, -1, -2, \ldots$ をみたし，$(\alpha, n) = \overbrace{\alpha \cdot (\alpha + 1) \cdots (\alpha + n - 1)}^{n}$ とする．この級数は $|z| < 1$ において絶対収束している．

Theorem 1

$$\mathrm{agm}(a, b) = \frac{a}{F\left(\frac{1}{2}, \frac{1}{2}, 1; 1 - \frac{b^2}{a^2}\right)}.$$

Maxima には, 超幾何級数が用意されている.

```
Agm(x,y):=bfloat(x/hypergeometric([1/2,1/2],[1],1-(y/x)^2));
Agm(sqrt(2),1);
Gagm;
Agm(sqrt(2),1)-Gagm;

agm(19,7);
Agm(19,7);

agm(218,122);
Agm(218,122);
```

Theorem 1 の証明に利用する事実:

Theorem 2 (不変原理) $m_1(x,y)$, $m_2(x,y)$ が $D = \{(x,y) \in \mathbb{R}^2 \mid x,y > 0\}$ 上の連続関数で,

$$x \neq y \Rightarrow \min(x,y) < m_1(x,y), m_2(x,y) < \max(x,y)$$

をみたすとする. 数列 $\{x_n\}$, $\{y_n\}$ を初項 $x_0 = x, y_0 = y$ を正数で与え, 漸化式

$$x_{n+1} = m_1(x_n, y_n), \quad y_{n+1} = m_2(x_n, y_n),$$

で定める. $\{x_n\}$, $\{y_n\}$ は $n \to \infty$ のとき同じ値に収束し, その極限値は D 上の連続関数 $\mu(x,y)$ で

$$(1) \qquad \mu(x,x) = x, \quad \mu(m_1(x,y), m_2(x,y)) = \mu(x,y)$$

をみたす関数を用いて, $\mu(x,y)$ で表示できる.

Proof. 同一極限の存在 $\lim_{n\to\infty} x_n = \lim_{n\to\infty} y_n = \alpha$ を仮定する．関数 μ の性質より

$$\begin{aligned}\mu(x_0, y_0) &= \mu(m_1(x_0, y_0), m_2(x_0, y_0)) = \mu(x_1, y_1)\\ &= \mu(m_1(x_1, y_1), m_2(x_1, y_1)) = \mu(x_2, y_2)\\ &= \qquad\qquad\qquad \cdots \qquad\qquad\qquad = \mu(x_n, y_n)\end{aligned}$$

$n \to \infty$ とすると $x_n, y_n \to \alpha$ であり, 関数 μ の連続性より

$$\lim_{n\to\infty} \mu(x_n, y_n) = \mu(\lim_{n\to\infty} x_n, \lim_{n\to\infty} y_n) = \mu(\alpha, \alpha) = \alpha$$

が得られる． □

Theorem 1 を証明するには, (1) をみたす関数をみつけてしまえばよい．

Theorem 3 (２次変換公式)

$$(1+z)^{2\alpha}F\Big(\alpha,\alpha-\beta+\frac{1}{2},\beta+\frac{1}{2};z^2\Big)=F\Big(\alpha,\beta,2\beta;\frac{4z}{(1+z)^2}\Big).$$

Maxima で成立しているか確認してみよう．

```
z0:2/11; al:%pi; be:sin(2);
sahen:bfloat((1+z0)^(2*al)
              *hypergeometric([al,al-be+1/2],[be+1/2],z0^2));
 uhen:bfloat(hypergeometric([al,be],[2*be],4*z0/(1+z0)^2));
sahen-uhen;
```

Proof. $F(\alpha,\beta,\gamma;z)$ は超幾何微分方程式 (the hypergeometric differential equaiton)

$$z(1-z)\frac{d^2}{dz^2}f(z)+(\gamma-(\alpha+\beta+1)z)\frac{d}{dz}f(z)-\alpha\beta f(z)=0$$

をみたす．この微分方程式を用いて，両辺が同じ微分方程式をみたし，$z=0$ での両辺の値が一致することで証明できる． □

Theorem 1 の証明.

不変原理より

$$\frac{x}{F(\frac{1}{2},\frac{1}{2},1;1-\frac{x^2}{x^2})}=x,$$

$$\frac{x}{F(\frac{1}{2},\frac{1}{2},1;1-\frac{y^2}{x^2})}=\frac{\frac{x+y}{2}}{F(\frac{1}{2},\frac{1}{2},1;1-\frac{\sqrt{xy}^2}{(\frac{x+y}{2})^2})}$$

をみたすことを示せばよい.

上の式は $F(\alpha,\beta,\gamma;0)=1$ より, 下の式は２次変換公式に $\alpha=\beta=\frac{1}{2}$ とした公式よりわかる. $\square$

不変原理の応用

正数 x, y に対して,

$$m_3(x, y) = \left(\frac{\frac{1}{x} + \frac{1}{y}}{2}\right)^{-1} = \frac{2xy}{x + y}$$

を x, y の調和平均 (the harmonic mean of x, y) という.

$$m_2(x, y)^2 - m_3(x, y)^2 = xy - \frac{4x^2y^2}{(x+y)^2} = \frac{xy(x-y)^2}{(x+y)^2} \geq 0$$

より,

$$m_1(x, y) \geq m_2(x, y) \geq m_3(x, y) > 0.$$

数列 $\{x_n\}$, $\{y_n\}$ を初項 $x_0 = x, y_0 = y$ を正数で与え, 漸化式

$$x_{n+1} = m_1(x_n, y_n), \quad y_{n+1} = m_3(x_n, y_n)$$

で定める. $n \to \infty$ のとき, $\{x_n\}$, $\{y_n\}$ は同じに収束する. その極限値を x, y の算術調和平均 (the arithmetic-harmonic mean of x, y) といい, $\mathrm{ahm}(x, y)$ で表す.

Theorem 4

$$\mathrm{ahm}(x,y) = m_2(x,y) = \sqrt{xy}.$$

Proof. $m_2(x,x) = \sqrt{x \cdot x} = x$ と

$$\sqrt{m_1(x,y) \cdot m_3(x,y)} = \sqrt{\frac{2xy}{x+y} \cdot \frac{x+y}{2}} = \sqrt{xy},$$

より，不変原理より $\mathrm{ahm}(x,y) = \sqrt{xy}$. □

```
m3(x,y):=2*x*y/(x+y);
ahm(x,y):=block([i,xx,yy],array(xx,10),array(yy,10),
xx[0]:x,yy[0]:y,
for i from 0 thru 9 do (xx[i+1]:bfloat(m1(xx[i],yy[i])),
 yy[i+1]:bfloat(m3(xx[i],yy[i])),
 print(i,xx[i],yy[i],xx[i]-yy[i])),
return(xx[10]));

ahm(3,1); bfloat(m2(3,1));
```

新しい２つの平均 m_4, m_5 を以下で定め，

$$m_4(x,y) = \sqrt{\frac{x+y}{2}\cdot x}, \quad m_5(x,y) = \sqrt{\frac{x+y}{2}\cdot y}$$

$\{x_n\}, \{y_n\}$ の初項を正数 $x_0 = x, y_0 = y$ で与え，以下の漸化式で定める:

$$x_{n+1} = m_4(x_n, y_n), \quad y_{n+1} = m_5(x_n, y_n).$$

Theorem 5 (Carlson [10]，1971年)

$$\lim_{n\to\infty} x_n = \lim_{n\to\infty} y_n = \sqrt{\frac{x^2-y^2}{2\log(x/y)}}.$$

Proof. $\mu(x,y) = \frac{x^2-y^2}{2\log(x/y)}$ とおき，不変原理で必要になる等式

$$\mu(x,x) = x, \quad \mu(m_4(x,y), m_5(x,y)) = \mu(x,y)$$

を Maxima で証明してみよう．

```
m4(x,y):=sqrt((x+y)*x/2); m5(x,y):=sqrt((x+y)*y/2);
Cm(x,y):=sqrt((x^2-y^2)/(2*(log(x/y))));
limit(Cm(x,y),y,x);
factor(expand(Cm(m4(x,y),m5(x,y))/Cm(x,y))^2);
```

完全に等式を簡略化してくれていないが，必要な等式が得られた． □

```
cm(x,y):=block([i,xx,yy],array(xx,50),array(yy,50),
xx[0]:x,yy[0]:y,
for i from 0 thru 49 do (xx[i+1]:bfloat(m4(xx[i],yy[i])),
 yy[i+1]:bfloat(m5(xx[i],yy[i])),
if i<10 then print(i,xx[i],yy[i],xx[i]-yy[i])),
return(xx[50]));

cm(2,1);
bfloat(Cm(2,1));
```

平均の反復操作は，いつでも収束が早いわけではない．

円周率 π の算術幾何平均による数値計算

$\{x_n\}$, $\{y_n\}$ を初項 $x_0 = \sqrt{2}$, $y_0 = 1$, 漸化式 $x_{n+1} = m_1(x_n, y_n)$, $y_{n+1} = m_2(x_n, y_n)$ により定義された数列とする．数列 $\{z_n\}$ を

$$z_n = \sqrt{x_n^2 - y_n^2} \quad \left(= \frac{x_{n-1} - y_{n-1}}{2} \ (n \geq 1) \right)$$

で定める．z_n は急速に 0 へ収束する．

Theorem 6 (文献 [8], 6.2節)

$$\pi = \frac{\mathrm{agm}(\sqrt{2}, 1)^2}{1 - \sum_{n=0}^{\infty} 2^{n-1} z_n^2}.$$

この定理が π の数値計算に有効であることを1976年に Brent B.C. と Salamin E. が指摘している．この原理，つまり 数列 $\frac{x_{n+1}^2}{1-\sum_{i=0}^{n} 2^{i-1} z_i^2}$ を用いて，1982年に金田，吉野，田村 は π を 16,000,000 桁まで計算して，当時の π の近似計算世界記録を樹立している．

```
array(xx,10); array(yy,10);
xx[0]:sqrt(2); yy[0]:1; ww:1/2;
for i from 0 thru 9 do
(xx[i+1]:bfloat(m1(xx[i],yy[i])),
 yy[i+1]:bfloat(m2(xx[i],yy[i])),
 ww:ww-2^(i)*bfloat(xx[i+1]^2-yy[i+1]^2),
 print(i+1,bfloat(xx[i+1]^2/ww),bfloat(%pi-xx[i+1]^2/ww)));
```

不変原理は，m 項の m 個の平均に対しても有効である．

３項の３平均 M_1, M_2, M_3 を以下で定める．

$$M_1(x,y,z) = \frac{x+y+z}{3},$$
$$M_2(x,y,z) = \frac{yz+zx+xy}{x+y+z},$$
$$M_3(x,y,z) = \frac{3xyz}{yz+zx+xy}.$$

$\{x_n\}, \{y_n\}, \{z_n\}$ の初項を正数 $x_0 = x, y_0 = y, z_0 = z$ で与え，一般項は以下の漸化式で定める:

$$x_{n+1} = M_1(x_n, y_n, z_n),$$
$$y_{n+1} = M_2(x_n, y_n, z_n),$$
$$z_{n+1} = M_3(x_n, y_n, z_n).$$

Theorem 7 $\lim_{n\to\infty} x_n = \lim_{n\to\infty} y_n = \lim_{n\to\infty} z_n = \sqrt[3]{xyz}.$

Proof. 不変原理の等式 $\sqrt[3]{x \cdot x \cdot x} = x$,

$$\sqrt[3]{M_1(x,y,z)M_2(x,y,z)M_3(x,y,z)}$$
$$= \sqrt[3]{\frac{x+y+z}{3} \cdot \frac{yz+zx+xy}{x+y+z} \cdot \frac{3xyz}{yz+zx+xy}} = \sqrt[3]{xyz}$$

をみたしているで，極限公式が得られる． □

```
M1(x,y,z):=(x+y+z)/3; M2(x,y,z):=(y*z+z*x+x*y)/(x+y+z);
M3(x,y,z):=3*x*y*z/(y*z+z*x+x*y);
ahm3(x,y,z):=block([i,xx,yy,zz],array(xx,10),array(yy,10),
array(zz,10),xx[0]:x,yy[0]:y,zz[0]:z,
for i from 0 thru 9 do (xx[i+1]:bfloat(M1(xx[i],yy[i],zz[i])),
 yy[i+1]:bfloat(M2(xx[i],yy[i],zz[i])),
 zz[i+1]:bfloat(M3(xx[i],yy[i],zz[i])),
 print(i,xx[i],yy[i],zz[i],xx[i]-zz[i])), return(xx[10]));
ahm3(2,1,1); bfloat(2^(1/3));
```

収束が早く 3 乗根を効率よく計算するアルゴリズムを与える．

算術幾何平均に関する研究史その３：

C.W. Borchardt は 1876年の論文 [9] において，4 項の 4 種平均

$$m_1(x) = \frac{x+y+z+w}{4}, \quad m_2(x) = \frac{\sqrt{xy}+\sqrt{zw}}{2},$$
$$m_3(x) = \frac{\sqrt{xz}+\sqrt{yw}}{2}, \qquad m_4(x) = \frac{\sqrt{xw}+\sqrt{yz}}{2},$$

を用いて，$\{x_n\}$，$\{y_n\}$，$\{z_n\}$，$\{w_n\}$ を初項を正数 $x_0 = x$，$y_0 = y$，$z_0 = z$，$w_0 = w$ で与え，漸化式

$$x_{n+1} = m_1(x_n, y_n, z_n, w_n), \quad y_{n+1} = m_2(x_n, y_n, z_n, w_n),$$
$$z_{n+1} = m_3(x_n, y_n, z_n, w_n), \quad w_{n+1} = m_4(x_n, y_n, z_n, w_n),$$

により定め，極限

$$\lim_{n\to\infty} x_n = \lim_{n\to\infty} y_n = \lim_{n\to\infty} z_n = \lim_{n\to\infty} w_n$$

の表示公式を与えた．その表示は，原始関数が初等関数で表示できない広義積分を成分とする 2×2 行列の行列式となっている．残念ながら Maxima では，その数値計算は容易にはできない．

４項の４種平均を算術平均は残して，残りの３つを４項を２項ごとに分けて，２項ごと算術平均とその結果に対する幾何平均で定義する．つまり，

$$m_1(x) = \frac{x+y+z+w}{4}, \qquad m_2(x) = \frac{\sqrt{(x+w)(y+z)}}{2},$$

$$m_3(x) = \frac{\sqrt{(x+z)(y+w)}}{2}, \quad m_4(x) = \frac{\sqrt{(x+y)(z+w)}}{2},$$

で定める．これらを用いて，４種の数列 $\{x_n\}$，$\{y_n\}$，$\{z_n\}$，$\{w_n\}$ を初項を正数 $x_0 = x$，$y_0 = y$，$z_0 = z$，$w_0 = w$ で与え，漸化式

$$\begin{aligned} x_{n+1} &= m_1(x_n, y_n, z_n, w_n), \quad y_{n+1} = m_2(x_n, y_n, z_n, w_n), \\ z_{n+1} &= m_3(x_n, y_n, z_n, w_n), \quad w_{n+1} = m_4(x_n, y_n, z_n, w_n), \end{aligned}$$

により定める．極限

$$\lim_{n\to\infty} x_n = \lim_{n\to\infty} y_n = \lim_{n\to\infty} z_n = \lim_{n\to\infty} w_n = \mathrm{agm}_4(x, y, z, w)$$

の表示公式を与えた．

3 変数の超幾何級数 Lauricella's F_D は

$$F_D(\alpha,\beta_1,\beta_2,\beta_3,\gamma;z_1,z_2,z_3)$$
$$=\sum_{n_1,n_2,n_3=0}^{\infty}\frac{(\alpha,n_1+n_2+n_3)(\beta_1,n_1)(\beta_2,n_2)(\beta_3,n_3)}{(\gamma,n_1+n_2+n_3)n_1!n_2!n_3!}z_1^{n_1}z_2^{n_2}z_3^{n_3}$$

で定義される.

Theorem 8 ([12])

$$\mathrm{agm}_4(x,y,z,w)=\frac{x}{F_D(\frac{1}{4},\frac{1}{4},\frac{1}{4},\frac{1}{4},1;1-\frac{y^2}{x^2},1-\frac{z^2}{x^2},1-\frac{w^2}{x^2})^2}$$

```
M1(x,y,z,w):=(x+y+z+w)/4;
M2(x,y,z,w):=sqrt((x+w)*(y+z))/2;
M3(x,y,z,w):=sqrt((x+z)*(y+w))/2;
M4(x,y,z,w):=sqrt((x+y)*(z+w))/2;
agm4(x,y,z,w):=block([aa,bb,cc,dd,i],
array(aa,10), array(bb,10), array(cc,10), array(dd,10),
aa[0]:x, bb[0]:y, cc[0]:z, dd[0]:w,
for i from 0 thru 9 do
(aa[i+1]:float(M1(aa[i],bb[i],cc[i],dd[i])),
 bb[i+1]:float(M2(aa[i],bb[i],cc[i],dd[i])),
 cc[i+1]:float(M3(aa[i],bb[i],cc[i],dd[i])),
 dd[i+1]:float(M4(aa[i],bb[i],cc[i],dd[i])),
 print(i,aa[i],bb[i],cc[i],dd[i])),
return(bb[10]));
```

```
PH(x,y):=prod((x+i),i,0,y-1);
PW(x,y):=if x=0 and y=0 then 1 else x^y;
F_D(x,y,z):=1+
float(sum(sum(sum(PH(1/4,n)
*PH(1/4,i1)*PH(1/4,i2)*PH(1/4,n-i1-i2)
/(n!*i1!*i2!*(n-i1-i2)!)*PW(x,i1)*PW(y,i2)*PW(z,n-i1-i2),
i1,0,n-i2),i2,0,n),n,1,30));
Agm4(x,y,z,w):=x/F_D(1-y^2/x^2,1-z^2/x^2,1-w^2/x^2)^2;
```

```
x0:4; y0:3; z0:2; w0:1;
agm4(x0,y0,z0,w0);
Agm4(x0,y0,z0,w0);

Agm4(M1(x0,y0,z0,w0),M2(x0,y0,z0,w0),
M3(x0,y0,z0,w0),M4(x0,y0,z0,w0));

x0:7; y0:5; z0:3; w0:2;
agm4(x0,y0,z0,w0);
Agm4(x0,y0,z0,w0);

Agm4(M1(x0,y0,z0,w0),M2(x0,y0,z0,w0),
M3(x0,y0,z0,w0),M4(x0,y0,z0,w0));
```

References

[1] 上見練太郎, 勝股脩, 加藤重雄, 久保田幸次, 神保秀一, 山口佳三 共著, 微分 改訂版, 共立出版, 東京, 2014.

[2] 上見練太郎, 勝股脩, 加藤重雄, 久保田幸次, 神保秀一, 山口佳三 共著, 積分 改訂版, 共立出版, 東京, 2014.

[3] 泉屋周一, 石川剛郎, 陳蘊剛, 上見練太郎, 三波篤郎, 西森敏之 共著, 行列と連立一次方程式, 共立出版, 東京, 1996.

[4] 泉屋周一, 石川剛郎, 陳蘊剛, 上見練太郎, 三波篤郎, 西森敏之 共著, 線形写像と固有値, 共立出版, 東京, 1996.

[5] 三宅 敏恒 著,
入門線形代数, 培風館, 東京, 1991.

[6] 三宅 敏恒 著,
入門微分積分, 培風館, 東京, 1992.

[7] 高木 貞治 著,
近世数学史談, 共立出版, 東京, 1987.

[8] 梅村 浩 著,
楕円関数論, 東京大学出版会, 東京, 2000.

[9] Borchardt C.W., Über das arithmetisch-geometrische Mittel aus vier Elementen, Berl. Monatsber, **53** (1876), 611-621.

[10] Carlson B.C., Algorithms involving arithmetic and geometric means, MAA Monthly, 78(5) (1971), 496-505.

[11] Gauss C.F. 日記, Gauss 全集第X巻 1, 485–574, Georg Olms Verlag, Hilchesheim-New York, 1981.

[12] Kato T. and Matsumoto K., The common limit of a quadruple sequence and the hypergeometric function F_D of three variables, Nagoya Math. J. **195** (2009), 113–124.